
Practicing Dynare

Francisco Barillas

New York University

Dept. of Economics∗†

Anmol Bhandari

New York University

Dept. of Economics

Riccardo Colacito

University of North Carolina

Dept. of Finance

Sagiri Kitao

University of Southern California

FBE Deptartment

Christian Matthes

New York University

Dept. of Economics

Thomas J. Sargent

New York University

Dept. of Economics

Yongseok Shin

Washington University

Dept. of Economics

December 4, 2010

Abstract

This paper teaches Dynare by applying it to approximate equilibria and estimate nine dynamic economic models.

We hope that it illustrates once again the efficacy of learning by doing.

1. Introduction

This paper describes nine examples or types of examples that illustrate how Dynare can approximate the solutions
of dynamic rational expectations models, simulate them, and estimate them by maximum likelihood and Bayesian
methods. Section 2 approximates and estimates a one-sector stochastic growth model. Section 3 approximates and
estimates a two-country stochastic growth model. Section 4 follows chapter 11 of Ljungqvist and Sargent (2004) in
studying the effects of foreseen fiscal policy in a non-stochastic growth model. Section 5 updates and extends these
examples to correspond to chapter 11 of Ljungqvist and Sargent (20XX). Section 6 estimates a rational expectations
model of hyperinflation originally formulated by Sargent (1977). Section 7 solves and estimates the permanent
income model of Hall (1988). This application is interesting, among other reasons, for the way that it illustrates how
Dynare can implement the ‘diffuse Kalman filter’ needed in situations in which an initial endogenous state variable
is unknown and the model has a unit root. Section 8 solves and estimates the Ryoo and Rosen (2004) rational
expectations model of a market for engineers. Section 9 estimates a model of consumption growth proposed by
Bansal and Yaron (2004). Section 10 solves and estimates the Hansen, Sargent, and Tallarini (1999) model of robust
permanent income and pricing. Section 11 solves and estimates the Bansal and Yaron (2004) of asset prices and
long-run risk in consumption and dividends. Appendix A tells how the reader can obtain the file examples.zip that
contain the *.mod and data files that we used to generate these examples.

Some of our examples bear eloquent witness to the technological improvements brought by Dynare. For example,
the maximum likelihood estimates in Sargent (1977) and Hansen, Sargent, and Tallarini (1999) were time-consuming
and painful to obtain originally. Dynare has reduced the time and the pain.

∗We thank Michele Juillard for helpful comments on an earlier draft.
†Authors’ emails: Barillas: fvb201@nyu.edu, Bhandari: apb296@nyu.edu, Colacito: riccardo.colacito@unc.edu, Kitao:

sk910@nyu.edu, Matthes, cm1518@nyu.edu, Sargent: ts43@nyu.edu, Shin: yshin@wustl.edu.

2. The Neoclassical growth model

2.1 The model

We study a widely used stochastic neoclassical growth model with leisure (see, for example, Cooley and Prescott
(1995)). A representative household’s problem is

max
{ct,lt}∞

t=0

E0

∞
∑

t=1

βt−1

(

cθ
t (1 − lt)

1−θ
)1−τ

1 − τ

subject to the resource constraint
ct + it = eztkα

t l1−α
t (1)

the law of motion for capital
kt+1 = it + (1 − δ)kt (2)

and the stochastic process for productivity
zt = ρzt−1 + sǫt (3)

with ǫt ∼ N(0, σ2).

The system of equations characterizing an equilibrium is comprised of equations (1), (2) and (3), the Euler
intertemporal condition

(

cθ
t (1 − lt)

1−θ
)1−τ

ct

= βEt

[

(

cθ
t+1(1 − lt+1)

1−θ
)1−τ

ct+1
(1 + αeztkα−1

t lαt − δ)

]

(4)

and an optimality condition for supply of labor

1 − θ

θ

ct

1 − lt
= (1 − α)eztkα

t l−α
t . (5)

Use (1) and (2) to get
kt+1 = eztkα

t l1−α
t − ct + (1 − δ)kt. (6)

An equilibrium is characterized by a system of four equations (3), (4), (5) and (6).

2.2 Calibration and approximation

2.2.1 Calibration

To estimate the policy function, parameters that enter the model are calibrated as in Table 1. For more on the
calibrations of the related parameters, please see, for example, Cooley and Prescott (1995).

2.2.2 Approximation

The following instructions implement this model in Dynare.

periods 1000;

var c k lab z;

varexo e;

parameters bet the del alp tau rho s;

bet = 0.987;

the = 0.357;

del = 0.012;

alp = 0.4;

tau = 2;

rho = 0.95;

s = 0.007;

model;

(c^the*(1-lab)^(1-the))^(1-tau)/c=bet*((c(+1)^the*(1-lab(+1))^(1-the))^(1-tau)/c(+1))*

(1+alp*exp(z(-1))*k(-1)^(alp-1)*lab^(1-alp)-del);

c=the/(1-the)*(1-alp)*exp(z(-1))*k(-1)^alp*lab^(-alp)*(1-lab);

k=exp(z(-1))*k(-1)^alp*lab^(1-alp)-c+(1-del)*k(-1);

z=rho*z(-1)+s*e;

end;

initval;

k = 1;

c = 1;

lab = 0.3;

z = 0;

e = 0;

end;

shocks;

var e;

stderr 1;

end;

steady;

stoch_simul(dr_algo=0,periods=1000); datasaver(’simudata’,[]);

Parameter Calibration
β 0.987
θ 0.357
δ 0.012
α 0.4
τ 2
ρ 0.95
s 0.007
σ 1

Table 1: Parameters’ calibration

Dynare displays the results reported in Table 2 that provide the coefficients of the approximated policy functions
and transition functions.

k z c lab
Constant 54.615371 0 0.960775 0.543716

k (-1) 0.983153 0 0.011727 -0.001668
z (-1) 1.978406 0.950000 0.317480 0.228716

e -0.000935 0.007000 0.000454 -0.000162
k (-1),k (-1) -0.000038 0 -0.000030 0.000011
z (-1),k (-1) 0.011128 0 0.002996 0.000762
z (-1),z (-1) 1.253721 0 0.096224 -0.046394

e,e -0.000003 0 0.000001 0
k (-1),e -0.000002 0 0.000002 0.000001
z (-1),e -0.001076 0 0.000430 -0.000056

Table 2: Policy and transition functions.

Figure 1: Impulse response function to a technology shock

2.3 Estimation

The simulated data series for consumption are used to estimate some unknown parameters of the model. Suppose we
would like to estimate the preference parameters θ and τ , and the stochastic process for productivity, summarized
by two parameters ρ and σ. The following code instructs Dynare to conduct the estimation.

periods 1000;

var c k lab z;

varexo e;

parameters bet del alp rho the tau s;

bet = 0.987;

the = 0.357;

del = 0.012;

alp = 0.4;

tau = 2;

rho = 0.95;

s = 0.007;

model;

(c^the*(1-lab)^(1-the))^(1-tau)/c=bet*((c(+1)^the*(1-lab(+1))^(1-the))^(1-tau)/c(+1))

*(1+alp*exp(z(-1))*k(-1)^(alp-1)*lab^(1-alp)-del);

c=the/(1-the)*(1-alp)*exp(z(-1))*k(-1)^alp*lab^(-alp)*(1-lab);

k=exp(z(-1))*k(-1)^alp*lab*(1-alp)-c+(1-del)*k(-1);

z=rho*z(-1)+s*e;

end;

initval;

k = 1;

c = 1;

lab = 0.3;

z = 0;

e = 0;

end;

shocks;

var e;

stderr 1;

end;

steady;

check;

estimated_params;

stderr e,inv_gamma_pdf, 0.95,inf;

rho,beta_pdf,0.93,0.02;

the,normal_pdf,0.3,0.05;

tau,normal_pdf,2.1,0.3;

end;

varobs c;

estimation(datafile=simudata,mh_replic=10000);

The priors used in estimation are as in Table 3. Table 4 is one of the outputs of Dynare which presents summary
statistics of posterior distribution.

Figure 2: Priors.

Figure 3: Posterior.

Parameter Distribution Mean Std.Dev.
ρ Beta 0.93 0.02
θ Normal 0.3 0.05
τ Normal 2.1 0.3
σ Inv. Gamma 0.95 inf.

Table 3: Priors

prior mean post. mean conf. interval prior dist prior std
ρ 0.930 0.9480 0.9319 0.9645 beta 0.0200
θ 0.300 0.3589 0.3540 0.3645 norm 0.0500
τ 2.100 2.0046 1.6532 2.3356 norm 0.3000
σ 0.950 1.0227 0.8321 1.2175 invg Inf

Table 4: Posterior.

3. International Business Cycle Model

We study a simplified version of the two-country production model in Kim and Kim (2003).

3.1 The model

Two countries are identical ex-ante and markets are complete. A solution to a Pareto planner’s problem is charac-
terized by the following equations

C1,t = C2,t

C−γ
1,t = βEtC

−γ
1,t+1

(

αA1,t+1K
α−1
1,t+1 + 1 − δ

)

C−γ
2,t = βEtC

−γ
2,t+1

(

αA2,t+1K
α−1
2,t+1 + 1 − δ

)

A1,tK
α
1,t + A2,tK

α
2,t = C1,t + C2,t + K1,t+1 − (1 − δ)K1,t + K2,t+1 − (1 − δ)K2,t

lnA1,t+1 = ρ lnA1,t + ǫ1,t+1

lnA2,t+1 = ρ lnA2,t + ǫ2,t+1

The first three equalities are the first-order conditions with regard to the consumption (Ci,t), and the fourth one
is the world resource constraint. The last two describe the law of motion for the technological shock (Ai,t). Ki,t

denotes the capital stock of country i at time t.

3.2 Calibration and approximation

3.2.1 Calibration

The following parameter values are used.

Parameter Calibration
β 0.98
δ 0.05
α 0.4
ρ 0.85

σǫ1 0.08
σǫ2 0.08

Table 5: Parameter Values

Dynare computes the policy function as a second-order approximation around the (log) steady-state. The .mod

Dynare file of the system is as follows.

periods 1000;

var c1 c2 k1 k2 a1 a2;

varexo e1 e2;

parameters gamma delta alpha beta rho;

gamma=2;

delta=.05;

alpha=.4;

beta=.98;

rho=.85;

model;

c1=c2;

exp(c1)^(-gamma) =...

beta*exp(c1(+1))^(-gamma)*(alpha*exp(a1(+1))*exp(k1)^(alpha-1)+1-delta);

exp(c2)^(-gamma) =...

beta*exp(c2(+1))^(-gamma)*(alpha*exp(a2(+1))*exp(k2)^(alpha-1)+1-delta);

exp(c1)+exp(c2)+exp(k1)-exp(k1(-1))*(1-delta)+exp(k2)-exp(k2(-1))*(1-delta)...

= exp(a1)*exp(k1(-1))^alpha+exp(a2)*exp(k2(-1))^alpha;

a1=rho*a1(-1)+e1; a2=rho*a2(-1)+e2; end;

initval;

k1=2.8;

k2=2.8;

c1=.8;

c2=.8;

a1=0;

a2=0;

e1=0;

e2=0;

end;

shocks;

var e1;

stderr .08;

var e2;

stderr .08;

end;

steady;

stoch_simul(dr_algo=0,periods=1000); datatomfile(’simudata’,[]);

The output is given as a matrix of coefficients for the policy function. From this matrix, for example, one can
construct the optimal consumption rule for Country 1:

ĉ1,t = 0.824 + 0.243(k̂1,t + k̂2,t) + 0.140(â1,t + â2,t) + 0.062(k̂2
1,t−1 + k̂2

2,t−1) − 0.110k̂1,tk̂2,t

− 0.015(â1,tk̂1,t + â2,tk̂2,t) − 0.032(â1,tk̂2,t + â2,tk̂1,t) + 0.044(â2
1,t + â2

2,t) − 0.040â1,tâ2,t,

where ŷ denotes the deviation of lnY from its log steady-state. Note that Dynare breaks âi,t into âi,t−1 and ǫi,t, so
the output from Dynare looks unnecessarily messy.

Dynare also produces the associated impulse response functions of the system. Figure 4 illustrate the response
of the system to a one-standard-deviation shock to ǫ1.

The last two lines of the code generates sample paths of the variables governed by the approximated policy
functions. The simulated process is saved as a .m file.

3.3 Estimation

Now assume that we do not know the true parameter values. We can estimate the unknown parameters from the
saved data, using the Bayesian econometrics function of Dynare.

We will estimate 4 parameters ρ, α, σǫ1 , and σǫ2 . Hence, we have to specify the prior distribution of each
parameter. In this exercise, ρ and α are given a normal prior, while the priors of σǫ1 and σǫ2 are inverse gamma
distributions. The Dynare code is as follows.

periods 1000;

var c1 c2 k1 k2 a1 a2;

varexo e1 e2;

parameters gamma delta alpha beta rho;

Figure 4: Impulse response function (ǫ1)

gamma=2;

delta=.05;

alpha=.4;

beta=.98;

rho=.85;

model;

c1=c2; exp(c1)^(-gamma) =...

beta*exp(c1(+1))^(-gamma)*(alpha*exp(a1(+1))*exp(k1)^(alpha-1)+1-delta);

exp(c2)^(-gamma) =...

beta*exp(c2(+1))^(-gamma)*(alpha*exp(a2(+1))*exp(k2)^(alpha-1)+1-delta);

exp(c1)+exp(c2)+exp(k1)-exp(k1(-1))*(1-delta)+exp(k2)-exp(k2(-1))*(1-delta)...

= exp(a1)*exp(k1(-1))^alpha+exp(a2)*exp(k2(-1))^alpha;

a1=rho*a1(-1)+e1; a2=rho*a2(-1)+e2; end;

initval;

k1=2.8;

k2=2.8;

c1=.8;

c2=.8;

a1=0;

a2=0;

e1=0;

e2=0;

end;

shocks;

var e1;

stderr .08;

var e2;

stderr .08;

end;

steady;

estimated_params;

rho, normal_pdf, .84,.05; %

alpha, normal_pdf, .38, .03;

stderr e1, inv_gamma_pdf, .078, inf;

stderr e2,inv_gamma_pdf,.082, inf;

end;

varobs c1 k2;

estimation(datafile=simudata,mh_replic=1000,mh_jscale=.5);

Given that there are two exogenous shock variables, we use two observables ln C1 and lnK2. The result of the
estimation is summarized in Figure 5, where the posterior distribution (in the darker line) of each parameter is
contrasted against the given prior distribution. The result can be compared to the true parameter values in Table 5.

Figure 5: Prior v. Posterior Distribution

4. Fiscal policies in the growth model

This section describes a suite of Dynare programs that replicate the transition experiments performed in chapter 11
of Ljungqvist and Sargent (2004).

4.1 The model

A deterministic growth model has inelastic labor supply, an exogenous stream of government expenditures, and
several kinds of distorting taxes. All variables are as described in chapter 11 of Ljungqvist and Sargent (2004). A
representative agent maximizes

∞
∑

t=0

βt c1−γ
t

1 − γ
, β ∈ (0, 1)

Feasible allocations satisfy
gt + ct + kt+1 ≤ Akα

t + (1 − δ) kt.

The household budget constraint is

∞
∑

t=0

{qt (1 + τct) + (1 − τit) qt [kt+1 − (1 − δ) kt]} ≤
∞
∑

t=0

{rt (1 − τkt) kt + wt}

and the government budget constraint is

∞
∑

t=0

qtgt ≤

∞
∑

t=0

{τctqtct − τitqt [kt+1 − (1 − δ) kt] + rtτktkt + wt}

The following conditions characterize an equilibrium:

ct = Akα
t + (1 − δ) kt − kt+1 − gt (7)

qt = βtc−γ
t / (1 + τct) (8)

rt/qt = Aαkα−1
t (9)

wt/qt = Akα
t − ktAαkα−1

t (10)

Rt+1 =
(1 + τct)

(1 + τct+1)

[

(1 + τit+1)

(1 + τit)
(1 − δ) +

(1 − τkt+1)

(1 − τit)
Aαkα−1

t+1

]

(11)

st/qt = (1 − τkt)Aαkα−1
t + (1 − δ) (12)

c−γ
t = βc−γ

t+1Rt+1 (13)

Because the other endogenous variables can be expressed as functions of kt and ct, in the programs below we use
only a subset of the equilibrium conditions to compute equilibrium paths.

4.2 Parameter values

We set the parameters and the baseline values of the exogenous variables at values set in chapter 11 of Ljungqvist
and Sargent (2004), as described in the following table are taken from RMT2.

Parameter/exogenous variable Values
β 0.95
δ 0.2
A 1
α 0.33
γ 2
g 0.2
τc 0
τi 0
τk 0

4.3 Transition experiments

4.3.1 A permanent increase in g (Figure 11.3.1).

The Dynare code below reproduces Figure 11.3.1. which looks at the impact of a permanent increase of 0.2 in g at
t=10.

// This program replicates figure 11.3.1 from chapter 11 of RMT2 by

// Ljungqvist and Sargent.

// y_ records the simulated endogenous variables in alphabetical order

// ys0_ records the initial steady state

// ys_ records the terminal steady state

// We check that these line up at the end points

// Note: y_ has ys0_ in the first column, ys_ in the last column, which explains

// why it has 102 elements.

// The sample of size 100 is in between.

// Warning: All endogenous variables dated time t are treated as jump variables

// in Dynare. So k in the program corresponds to k_{t+1}

// and the same timing holds for the taxes.

//Declares the endogenous variables;

var c k;

//declares the exogenous variables

//investment tax credit, consumption tax, capital tax, government spending

varexo taui tauc tauk g;

parameters bet gam del alpha A;

bet=.95; // discount factor

gam=2; // CRRA parameter

del=.2; // depreciation rate

alpha=.33; // capital’s share

A=1; // productivity

// Alignment convention:

// g tauc taui tauk are now columns of ex_. Because of a bad design decision

// the date of ex_(1,:) doesn’t necessarily match the date in y_.

// Whether they match depends on the number of lag periods in endogenous versus

// exogenous variables.

// In this example they match because tauc(-1) and taui(-1) enter the model.

// These decisions and the timing conventions mean that y_(:,1) records the

// initial steady state, while y_(:,102) records the terminal steady state

// values. For j > 2, y_(:,j) records [c(j-1) .. k(j-1) .. G(j-1)] where k(j-1)

// means end of period capital in period j-1, which equals k(j) in chapter 11

// notation. Note that the jump in G occurs in y_(;,11), which confirms this

// timing. The jump occurs now in ex_(11,1)

model;

// equation 11.3.8.a

k=A*k(-1)^alpha+(1-del)*k(-1)-c-g;

// equation 11.3.8e + 11.3.8.g

c^(-gam)=

bet*(c(+1)^(-gam))*((1+tauc(-1))/(1+tauc))*((1-taui)*(1-del)/(1-taui(-1))+

((1-tauk)/(1-taui(-1)))*alpha*A*k(-1)^(alpha-1));

end;

initval; k=1.5; c=0.6; g = 0.2; tauc = 0; taui = 0; tauk = 0; end;

steady; // put this in if you want to start from the initial

// steady state, comment it out to start from the indicated values

endval; // The following values determine the new steady state

// after the shocks.

k=1.5; c=0.4; g =.4; tauc =0; taui =0; tauk =0; end;

steady; // We use steady again and the enval provided are initial

// guesses for Dynare to compute the ss.

// The following lines produce a g sequence with a once and for all jump in g shocks

// we use shocks to undo that for the first 9 periods and leave g at

// it’s initial value of 0.2

var g; periods 1:9; values 0.2; end;

// now solve the model

simul(periods=100);

// Note: y_ records the simulated endogenous variables in alphabetical order

// ys0_ records the initial steady state

// ys_ records the terminal steady state

// check that these line up at the end points

y_(:,1) -ys0_(:) y_(:,102) - ys_(:)

// Compute the initial steady state for consumption to later do the plots.

co=ys0_(var_index(’c’)); ko = ys0_(var_index(’k’));

// g is in ex_(:,1) since it is stored in alphabetical order

go = ex_(1,1)

// The following equation compute the other endogenous variables use in the

// plots below. Since they are function of capital and consumption, so we can

// compute them from the solved model above.

// These equations were taken from page 333 of RMT2

rbig0=1/bet;

rbig=y_(var_index(’c’),2:101).^(-gam)./(bet*y_(var_index(’c’),3:102).^(-gam));

rq0=alpha*A*ko^(alpha-1);

rq=alpha*A*y_(var_index(’k’),1:100).^(alpha-1);

wq0=A*ko^alpha-ko*alpha*A*ko^(alpha-1);

wq=A*y_(var_index(’k’),1:100).^alpha-y_(var_index(’k’),1:100).*alpha*A.*...

y_(var_index(’k’),1:100).^(alpha-1);

sq0=(1-ex_(1,4))*A*alpha*ko^(alpha-1)+(1-del);

sq=(1-ex_(1:100,4)’)*A*alpha.*y_(var_index(’k’),1:100).^(alpha-1)+(1-del);

//Now we plot the responses of the endogenous variables to the shock.

figure subplot(2,3,1) plot([ko*ones(100,1)

y_(var_index(’k’),1:100)’ ]) // note the timing: we lag capital to

correct for syntax title(’k’) subplot(2,3,2) plot([co*ones(100,1)

y_(var_index(’c’),2:101)’ ]) title(’c’) subplot(2,3,3)

plot([rbig0*ones(100,1) rbig’ ]) title(’R’) subplot(2,3,4)

plot([wq0*ones(100,1) wq’ ]) title(’w/q’) subplot(2,3,5)

plot([sq0*ones(100,1) sq’ ]) title(’s/q’) subplot(2,3,6)

plot([rq0*ones(100,1) rq’ ]) title(’r/q’)

Figure 6: A permanent increase of g (Figure 11.3.1).

The following figure shows the path of the endogenous variables along the transition to the new steady state.
Next we show the other transition experiments and only the parts of the code that need to be changed for each

experiment.

4.3.2 A permanent increase in τc (Figure 11.3.2).

A permanent increase of τc at t=10 of 20 per cent. (Figure 11.3.2)

initval;

k=1.5;

c=0.6;

g = 0.2;

tauc = 0;

taui = 0;

tauk = 0;

end;

steady;

endval;

k=1.5;

c=0.6;

g = 0.2;

tauc =0.2;

taui =0;

tauk =0;

end;

Figure 7: A permanent increase of τc (Figure 11.3.2).

steady;

shocks;

var tauc;

periods 1:9;

values 0;

end;

4.3.3 A permanent increase in τi (Figure 11.5.1).

A permanent increase of τi at t=10 of 20 percent. (Figure 11.5.1)

initval;

k=1.5;

c=0.6;

g = 0.2;

tauc = 0;

taui = 0;

tauk = 0;

end;

steady;

endval;

k=1.5;

c=0.6;

Figure 8: A permanent increase of τi (Figure 11.5.1).

g =0.2;

tauc =0;

taui =0.20;

tauk =0;

end;

steady;

shocks;

var taui;

periods 1:9;

values 0;

end;

4.3.4 A permanent increase in τk (Figure 11.5.2).

A permanent increase of τk at t=10 of 20 percent. (Figure 11.5.2)

initval;

k=1.5;

c=0.6;

g = 0.2;

tauc = 0;

taui = 0;

tauk = 0;

end;

steady;

Figure 9: A permanent increase of τk (Figure 11.5.2).

endval;

k=1.5;

c=0.6;

g =0.2;

tauc =0;

taui =0;

tauk = 0.2;

end;

steady;

shocks;

var tauk;

periods 1:9;

values 0;

end;

4.3.5 One time increase in g (Figure 11.7.1).

A one time pulse of g at t=10 of 0.2. (Figure 11.7.1)

initval;

k=1.5;

c=0.6;

g = 0.2;

tauc = 0;

Figure 10: One time increase of g (Figure 11.7.1).

taui = 0;

tauk = 0;

end;

steady;

endval;

k=1.5;

c=0.6;

g = 0.2;

tauc =0;

taui =0;

tauk =0;

end;

steady;

shocks;

var g;

periods 10;

values 0.4;

end;

4.3.6 One time increase in τi (Figure 11.7.2).

A one time pulse of τi at t=10 of 20 percent. (Figure 11.7.2)

initval;

Figure 11: One time increase of τi (Figure 11.7.2).

k=1.5;

c=0.6;

g = 0.2;

tauc = 0;

taui = 0;

tauk = 0;

end;

steady;

endval;

k=1.5;

c=0.6;

g =0.2;

tauc =0;

taui =0;

tauk =0;

end;

steady;

shocks;

var taui;

periods 10;

values 0.2;

end;

5. Fiscal policies in the growth model

This section describes a suite of Dynare programs that replicate the transition experiments performed in chapter 11
of Ljungqvist and Sargent (20XX).

5.1 The model with inelastic labor supply

A deterministic growth model has inelastic labor supply, an exogenous stream of government expenditures, and
several kinds of distorting taxes. All variables are as described in chapter 11 of Ljungqvist and Sargent (20XX). A
representative agent maximizes

∞
∑

t=0

βt c1−γ
t

1 − γ
, β ∈ (0, 1)

Feasible allocations satisfy

gt + ct + kt+1 ≤ Akα
t + (1 − δ) kt.

The household budget constraint is

∞
∑

t=0

{qtct (1 + τct) + qt [kt+1 − (1 − δ) kt]} ≤

∞
∑

t=0

qt {ηt (1 − τkt) kt + wt}

and the government budget constraint is

∞
∑

t=0

qtgt ≤

∞
∑

t=0

qt {τctct + τktηtkt}

The following conditions characterize an equilibrium:

ct = Akα
t + (1 − δ) kt − kt+1 − gt (14)

qt = βtc−γ
t / (1 + τct) (15)

ηt = Aαkα−1
t (16)

wt = Akα
t − ktAαkα−1

t (17)

R̄t+1 =
(1 + τct)

(1 + τct+1)

[

(1 − δ) + (1 − τkt+1)Aαkα−1
t+1

]

(18)

c−γ
t = βc−γ

t+1R̄t+1 (19)

Because the other endogenous variables can be expressed as functions of kt and ct, in the programs below we use
only a subset of the equilibrium conditions to compute equilibrium paths.

5.2 Parameter values

We set the parameters and the baseline values of the exogenous variables at values set in chapter 11 of Ljungqvist
and Sargent (20XX), as described in the following table are taken from RMT3.

Parameter/exogenous variable Values
β 0.95
δ 0.2
A 1
α 0.33
γ 2
g 0.2
τc 0
τk 0

5.3 Transition experiments

5.3.1 Foreseen once-and-for-all increase in g (Figure 11.6.1).

The Dynare code below reproduces Figure 11.6.1. which looks at the impact of a foreseen once-for-and-all increase
of 0.2 in g at t = 10.

%-----------------------------------------------------------------------------------------------

% Experiment : Permenant increase in g at t=10

% Note:Following the discussion in the text t = 0 is the first period of the new (simulated) path

%-----------------------------------------------------------------------------------------------

// This program replicates figure 11.6.1 from chapter 11 of RMT3 by Ljungqvist and Sargent

// Dynare records the endogenous variables with the following convention.

//Say N is the number of simulations(sample)

//Index 1 : Initial values (steady sate)

//Index 2 to N+1 : N simulated values

//Index N+2 : Terminal Value (Steady State)

// Warning: we align c, k, and the taxes to exploit the dynare syntax.

//In Dynare the timing of the variable reflects the date when the variable is decided.

//For instance the capital stock for time ’t’ is decided in time ’t-1’(end of period).

//So a statement like k(t+1) = i(t) + (1-del)*k(t) would translate to

// k(t) = i(t) +(1-del)*k(t-1) in the code.

%-----------------------------------------------------------------------------------------------

% 1. Defining variables

%-----------------------------------------------------------------------------------------------

// Declares the endogenous variables consumption (’c’) capital stock (’k’);

var c k;

// declares the exogenous variables consumption tax (’tauc’),

// capital tax(’tauk’), government spending(’g’)

varexo tauc tauk g;

parameters bet gam del alpha A;

%-----------------------------------------------------------------------------------------------

% 2. Calibration and alignment convention

%-----------------------------------------------------------------------------------------------

bet=.95; // discount factor

gam=2; // CRRA parameter

del=.2; // depreciation rate

alpha=.33; // capital’s share

A=1; // productivity

// Alignment convention:

// g tauc tauk are now columns of ex_. Because of a bad design decision the date of ex_(1,:)

// doesn’t necessarily match the date in endogenous variables. Whether they match depends on

// the number of lag periods in endogenous versus exogenous variables.

// These decisions and the timing conventions mean that

// k(1) records the initial steady state, while k(102) records the terminal steady state values.

// For j > 1, k(j) records the variables for j-1th simulation where the capital stock decision

// taken in j-1th simulation i.e stock at the beginning of period j.

// The variable ex_ also follows a different timing convention i.e

// ex_(j,:) records the value of exogenous variables in the jth simulation.

// The jump in the government policy is reflected in ex_(11,1) for instance.

%-----------------------------------------------------------------------------------------------

% 3. Model

%-----------------------------------------------------------------------------------------------

model;

// equation 11.3.8.a

k=A*k(-1)^alpha+(1-del)*k(-1)-c-g;

// equation 11.3.8e + 11.3.8.g

c^(-gam)= bet*(c(+1)^(-gam))*((1+tauc)/(1+tauc(+1)))*((1-del) + (1-tauk(+1))*alpha*A*k^(alpha-1));

end;

%-----------------------------------------------------------------------------------------------

% 4. Computation

%-----------------------------------------------------------------------------------------------

initval;

k=1.5;

c=0.6;

g = .2;

tauc = 0;

tauk = 0;

end;

steady;

// put this in if you want to start from the initial steady state,comment it out to start

// from the indicated values

// The following values determine the new steady state after the shocks.

endval;

k=1.5;

c=0.6;

g =.4;

tauc =0;

tauk =0;

end;

steady;

// We use ’steady’ again and the endval provided are initial guesses for dynare to compute the ss.

// The following lines produce a g sequence with a once and for all jump in g

// we use ’shocks’ to undo that for the first 10 periods (t=0 until t=9)and leave g at

// it’s initial value of 0

// Note : period j refers to the value in the jth simulation

shocks;

var g;

periods 1:10;

values 0.2;

end;

// now solve the model

simul(periods=100);

// Compute the initial steady state for consumption to later do the plots.

c0=c(1);

k0 = k(1);

// g is in ex_(:,1) since it is stored in alphabetical order

g0 = ex_(1,1)

%-----------------------------------------------------------------------------------------------

% 5. Graphs and plots for other endogenous variables

%-----------------------------------------------------------------------------------------------

// Let N be the periods to plot

N=40;

// The following equation compute the other endogenous variables use in the plots below

// Since they are function of capital and consumption, so we can compute them from the solved

// model above.

// These equations were taken from page 371 of RMT3

rbig0=1/bet;

rbig=c(2:101).^(-gam)./(bet*c(3:102).^(-gam));

nq0=alpha*A*k0^(alpha-1);

nq=alpha*A*k(1:100).^(alpha-1);

wq0=A*k0^alpha-k0*alpha*A*k0^(alpha-1);

wq=A*k(1:100).^alpha-k(1:100).*alpha*A.*k(1:100).^(alpha-1);

// Now we plot the responses of the endogenous variables to the shock.

x=0:N-1;

figure(1)

// subplot for capital ’k’

subplot(2,3,1)

plot(x,[k0*ones(N,1)],’--k’, x,k(1:N),’k’,’LineWidth’,1.5)

// note the timing: we lag capital to correct for syntax

title(’k’,’Fontsize’,12)

set(gca,’Fontsize’,12)

// subplot for consumption ’c’

subplot(2,3,2)

plot(x,[c0*ones(N,1)],’--k’, x,c(2:N+1),’k’,’LineWidth’,1.5)

title(’c’,’Fontsize’,12)

set(gca,’Fontsize’,12)

// subplot for cost of capital ’R_bar’

subplot(2,3,3)

plot(x,[rbig0*ones(N,1)],’--k’, x,rbig(1:N),’k’,’LineWidth’,1.5)

title(’$\overline{R}$’,’interpreter’, ’latex’,’Fontsize’,12)

set(gca,’Fontsize’,12)

// subplot for rental rate ’eta’

subplot(2,3,4)

plot(x,[nq0*ones(N,1)],’--k’, x,nq(1:N),’k’,’LineWidth’,1.5)

title(’\eta’,’Fontsize’,12)

set(gca,’Fontsize’,12)

// subplot for the experiment proposed

subplot(2,3,5)

plot([0:9],ex_(1:10,1),’k’,’LineWidth’,1.5);

hold on;

plot([10:N-1],ex_(11:N,1),’k’,’LineWidth’,1.5);

hold on;

plot(x,[g0*ones(N,1)],’--k’,’LineWidth’,1.5)

title(’g’,’Fontsize’,12)

axis([0 N -.1 .5])

set(gca,’Fontsize’,12)

print -depsc fig_g.eps

The following figure shows the path of the endogenous variables along the transition to the new steady state.
Next we show the other transition experiments and only the parts of the code that need to be changed for each

experiment.

5.3.2 Foreseen once-and-for-all increase in τc (Figure 11.6.4).

A foreseen once-and-for-all increase in τc at t = 10 of 20 per cent.(Figure 11.6.4)

initval;

k=1.5;

c=0.6;

g = 0.2;

tauc = 0;

tauk = 0;

end;

steady;

endval;

k=1.5;

c=0.6;

g = 0.2;

tauc =0.2;

tauk =0;

end;

steady;

shocks;

Figure 12: Foreseen once-and-for-all increase in g at t = 10 (Figure 11.6.1).

var tauc;

periods 1:10;

values 0;

end;

5.3.3 Foreseen once-and-for-all increase in τk (Figure 11.6.5).

A foreseen once-and-for-all increase in τk at t = 10 of 20 percent. (Figure 11.6.5)

initval;

k=1.5;

c=0.6;

g = 0.2;

tauc = 0;

tauk = 0;

end;

steady;

endval;

k=1.5;

c=0.6;

g =0.2;

tauc =0;

tauk = 0.2;

end;

Figure 13: Foreseen once-and-for-all increase in τc at t = 10 (Figure 11.6.4).

steady;

shocks;

var tauk;

periods 1:10;

values 0;

end;

5.3.4 Foreseen one-time pulse increase in g (Figure 11.6.6).

A foreseen one-time pulse of g at t = 10 of 0.2. (Figure 11.6.6)

initval;

k=1.5;

c=0.6;

g = 0.2;

tauc = 0;

tauk = 0;

end;

steady;

endval;

k=1.5;

c=0.6;

g = 0.2;

tauc =0;

Figure 14: Foreseen once-and-for-all foreseen increase in τk at t = 10 (Figure 11.6.5).

tauk =0;

end;

steady;

shocks;

var g;

periods 11;

values 0.4;

end;

5.3.5 Foreseen once-and-for-all increase g at t = 10 for two economies (Figure 11.6.2).

Responses to a foreseen increase in g at t = 10 for two economies, the original economy with γ = 2 and an otherwise
identical economy with γ = .2 (Figure 11.6.2).

//values for gamma = 2

kgamma(:,1)=k;

cgamma(:,1)=c;

rbig(:,1)=cgamma(2:101,1).^(-gam)./(bet*cgamma(3:102,1).^(-gam));

//values for gamma=.2

gam=.2;

simul(periods=100);

Figure 15: Foreseen one-time pulse increase in g at t = 10 (Figure 11.6.6).

kgamma(:,2)=k;

cgamma(:,2)=c;

rbig(:,2)=cgamma(2:101,2).^(-gam)./(bet*cgamma(3:102,2).^(-gam));

%-----------------------------------------------------------------------------------------------

% 5. Graphs and plots for other endogenous variables

%-----------------------------------------------------------------------------------------------

N=40;

rbig0=1/bet;

nq0=alpha*A*k0^(alpha-1);

nq=alpha*A*kgamma(1:100,:).^(alpha-1);

wq0=A*k0^alpha-k0*alpha*A*k0^(alpha-1);

wq=A*kgamma(1:100,:).^alpha-kgamma(1:100,:).*alpha*A.*kgamma(1:100,:).^(alpha-1);

x=0:N-1;

figure(1)

// subplot for capital ’k’

subplot(2,3,1)

plot(x,[k0*ones(N,1)],’--k’, x,kgamma(1:N,1),’k’,x,kgamma(1:N,2),’-.k’,’LineWidth’,1.5) // note the timing: we

title(’k’,’Fontsize’,12)

set(gca,’Fontsize’,12)

// subplot for consumption ’c’

subplot(2,3,2)

plot(x,[c0*ones(N,1)],’--k’, x,cgamma(2:N+1,1),’k’,x,cgamma(2:N+1,2),’-.k’,’LineWidth’,1.5)

title(’c’,’Fontsize’,12)

set(gca,’Fontsize’,12)

// subplot for cost of capital ’R_bar’

subplot(2,3,3)

plot(x,[rbig0*ones(N,1)],’--k’, x,rbig(1:N,1),’k’,x,rbig(1:N,2),’-.k’,’LineWidth’,1.5)

title(’$\overline{R}$’,’interpreter’, ’latex’,’Fontsize’,12)

set(gca,’Fontsize’,12)

// subplot for rental rate ’eta’

subplot(2,3,4)

plot(x,[nq0*ones(N,1)],’--k’, x,nq(1:N,1),’k’,x,nq(1:N,2),’-.k’,’LineWidth’,1.5)

title(’\eta’,’Fontsize’,12)

set(gca,’Fontsize’,12)

// subplot for the experiment proposed

subplot(2,3,5)

plot([0:9],ex_(1:10,1),’k’,’LineWidth’,1.5);

hold on;

plot([10:N-1],ex_(11:N,1),’k’,’LineWidth’,1.5);

hold on;

plot(x,[g0*ones(N,1)],’--k’,’LineWidth’,1.5)

title(’g’,’Fontsize’,12)

axis([0 N -.1 .5])

set(gca,’Fontsize’,12)

5.3.6 Impact of an foreseen once-and-for-all increase in g on yield curves (Figure 11.6.3).

Impact of an foreseen once-and-for-all increase in g at t = 10 on yield curves. For this section we use some additional
formulas

qt =
βtc−γ

t

c−γ
0

(20)

yi,t = −
1

i
log

(

qt+i

qt

)

(21)

%-----------------------------------------------------------------------------------------------

% 5. Graphs and plots for other endogenous variables

%-----------------------------------------------------------------------------------------------

//Let N be the periods to plot

N=40;

rbig0=1/bet;

rbig=c(2:101).^(-gam)./(bet*c(3:102).^(-gam));

nq0=alpha*A*k0^(alpha-1);

nq=alpha*A*k(1:100).^(alpha-1);

wq0=A*k0^alpha-k0*alpha*A*k0^(alpha-1);

wq=A*k(1:100).^alpha-k(1:100).*alpha*A.*k(1:100).^(alpha-1);

Figure 16: Response to foreseen once-and-for-all increase in g at t = 10.The solid line is for γ = 2, while the
dashed-dotted line is for γ = .2 (Figure 11.6.2).

// computing the price system : q and steady state price system qs(t)=beta^t

//q(2) is normalized to 1

//qs(1) is normalized to 1

for n=1:102

q(n)=bet^n*c(n).^(-gam)/(bet^2*c(2)^(-gam));

qs(n)=bet^n*c(1).^(-gam)/(bet*c(1)^(-gam));

end

// computing the short term interest rates

rsmall=-log(q(3:102)./q(2:101));

rsmall0=rbig0-1;

// computing the term structure for n=1 to 40 for all periods

for t=1:62

for i=1:40

y(i,t)=log(q(t+i)/q(t))/-i;

end

end

//Now we plot the yield curves and other responses of the endogenous variables to the shock.

figure(1)

x=0:N-1;

// subplot for consumption ’c’

subplot(2,3,1)

plot(x,c0*ones(N,1),’--k’, x,c(2:N+1),’k’,’LineWidth’,1.5)

title(’c’,’Fontsize’,12)

set(gca,’Fontsize’,12)

// subplot for consumption ’q’

subplot(2,3,2)

plot(x,qs(1:N),’--k’, x,q(2:N+1),’k’,’LineWidth’,1.5)

title(’q’,’Fontsize’,12)

set(gca,’Fontsize’,12)

// subplot for rate of interest ’r’

subplot(2,3,3)

plot(x,[rsmall0*ones(N,1)],’--k’, x,rsmall(1:N),’k’,’LineWidth’,1.5)

title(’r’,’Fontsize’,12)

set(gca,’Fontsize’,12)

// subplot for yield curves at t=0,t=10,t=60

subplot(2,3,4)

plot(x,y(:,2),’k’, x,y(:,11),’-.k’,x,y(:,61),’--k’,’LineWidth’,1.5)

title(’’,’Fontsize’,12)

set(gca,’Fontsize’,12)

axis fill;

Figure 17: Response to foreseen once-and-for-all increase in g at t = 10. Yield curves for t = 0 (solid line), t = 10
(dashdotted line) and t = 60 (dashed line); term to maturity is on the x axis for the yield curve, time for the other
panels (Figure 11.6.3).

// subplot for the experiment proposed

subplot(2,3,5)

plot([0:9],ex_(1:10,1),’k’,’LineWidth’,1.5);

hold on;

plot([10:N-1],ex_(11:N,1),’k’,’LineWidth’,1.5);

hold on;

plot(x,[g0*ones(N,1)],’--k’,’LineWidth’,1.5)

title(’g’,’Fontsize’,12)

axis([0 N -.1 .5])

set(gca,’Fontsize’,12)

5.4 The model with elastic labor supply

In this section we modify the previous model by allowing for elastic labor supply. Now the representative agent
maximizes

∞
∑

t=0

βt log c + B(1 − n), β ∈ (0, 1)

where B is substantially greater than 1 to assure an interior solution n ∈ (0, 1) for labor supply.
Feasible allocations satisfy

gt + ct + kt+1 ≤ Akα
t n1−α

t + (1 − δ) kt.

The household budget constraint is

∞
∑

t=0

{qt (1 + τct) + qt [kt+1 − (1 − δ) kt]} ≤

∞
∑

t=0

qt {ηt (1 − τkt) kt + wt (1 − τnt)}

and the government budget constraint is

∞
∑

t=0

qtgt ≤

∞
∑

t=0

qt {τctct + τktηtkt + τntntwt}

The following conditions characterize the endogenous variables in the model equilibrium:

ct = Akα
t n1−α

t + (1 − δ) kt − kt+1 − gt (22)

c−1
t = βc−1

t+1

(

1 + τct

1 + τct+1

)

[

(1 − δ) + (1 − τkt+1)Aαkα−1
t+1 n1−α

t+1

]

(23)

Bct(1 + τct) = (1 − τnt)
[

(1 − α)Akα
t n−α

t

]

(24)

5.4.1 Elastic Labor supply : Unforseen once-and-for-all increase in g at t=0 (Figure 11.9.1).

An unforeseen once-and-for-all increase in g to .2 at t = 0 (Figure 11.9.1).

%-----------------------------------------------------------------------------------------------

% 1. Defining variables

%-----------------------------------------------------------------------------------------------

//Declares the endogenous variables consumption (’c’) capital stock (’k’);

var c k n;

//declares the exogenous variables

// consumption tax (’tauc’), capital tax(’tauk’), labor tax(’taun’) government spending(’g’)

varexo tauc tauk taun g;

parameters bet gam del alpha A B;

%-----------------------------------------------------------------------------------------------

% 2. Calibration and alignment convention

%-----------------------------------------------------------------------------------------------

bet=.95; // discount factor

gam=1; // CRRA parameter

del=.2; // depreciation rate

alpha=.33; // capital’s share

A=1; // productivity

B=3; // coefficient on leisure

%-----------------------------------------------------------------------------------------------

% 3. Model

%-----------------------------------------------------------------------------------------------

model;

//Feasibility

k=A*k(-1)^alpha*n^(1-alpha)+(1-del)*k(-1)-c-g;

//Euler equation

c^(-gam)= bet*(c(+1)^(-gam))*((1+tauc)/(1+tauc(+1)))*((1-del) + (1-tauk(+1))

*alpha*A*k^(alpha-1)*n(+1)^(1-alpha));

//Consumption leisure choice

B/c^(-gam)=(1-taun)*((1-alpha)*A*k(-1)^(alpha)*n^(-alpha))*(1+tauc)^-1;

end;

%-----------------------------------------------------------------------------------------------

% 4. Computation

%-----------------------------------------------------------------------------------------------

initval;

k=1.5;

c=0.6;

g = .2;

n=1.02;

tauc = 0;

tauk = 0;

taun=0;

end;

steady;

endval;

k=1.5;

c=0.6;

g =.4;

n=1.02;

tauc =0;

tauk =0;

taun=0;

end;

steady;

simul(periods=100);

c0=c(1);

k0 = k(1);

n0=n(1);

g0 = .2;

5.4.2 Elastic Labor supply : Unforseen once-and-for-all increase in τn at t=0 (Figure 11.9.2).

An unforeseen once-and-for-all increase in τn to 20 per cent at t = 0 (Figure 11.9.2).

initval;

Figure 18: Elastic labor supply: response to unforeseen increase in g at t = 0 (Figure 11.9.1).

k=1.5;

c=0.6;

g = .2;

n=1.02;

tauc = 0;

tauk = 0;

taun=0;

end;

steady;

endval;

k=1.5;

c=0.6;

g =.2;

n=1.02;

tauc =0;

tauk =0;

taun=0.2;

end;

steady;

Figure 19: Elastic labor supply: response to unforeseen increase in τn at t = 0 (Figure 11.9.2).

5.4.3 Elastic Labor supply : Foreseen once-and-for-all increase in τn at t=10 (Figure 11.9.3).

A foreseen once-and-for-all increase in τn to 20 per cent at t = 10 (Figure 11.9.3).

initval;

k=1.5;

c=0.6;

g = .2;

n=1.02;

tauc = 0;

tauk = 0;

taun=0;

end;

steady;

endval;

k=1.5;

c=0.6;

g =.2;

n=1.02;

tauc =0;

tauk =0;

taun=0.2;

end;

steady;

Figure 20: Elastic labor supply: response to foreseen increase in τn at t = 10 (Figure 11.9.3).

shocks;

var taun;

periods 1:10;

values 0;

end;

6. The demand for money during hyperinflations

Sargent (1977) specified and estimated a joint stochastic process for money growth and inflation that was designed
to make the adaptive expectations model of Cagan (1956) consistent with rational expectations. The model has the
form

mt − pt = απt + ut (25)

πt =
1 − λ

1 − λL
(pt − pt−1) =

1 − λ

1 − λL
xt (26)

µt = mt − mt−1 (27)

with α < 0, m being the log of money supply, π the expectation of inflation, p the log of the price level and u an
erro term whose specification is described below. The hypothesis of rational expectations imposes

πt = Etxt+1. (28)

Sargent (1977) reverse engineered a process for xt such that Cagan’s adaptive expectations scheme is equal to rational
expectations. The following conditions ensure the rationality of Cagan’s expectations scheme

ut = ut−1 + ηt (29)

µt =
1 − λ

1 − λL
xt + εt (30)

η and ε are both serially uncorrelated random error terms with mean zero and variances σ2
η and σ2

ε , respectively. For
estimating the model, it is convenient to represent it in the following form

(

xt

µt

)

=

(

1 0
1 − λ λ

) (

xt−1

µt−1

)

+

(

a1t

a2t

)

− λI

(

a1t−1

a2t−1

)

(31)

The a terms are linked to the random terms above via
(

a1t

a2t

)

=

(

1 −1
1 + α(1 − λ) −(1 − λ)

)

(λ + (1 − λ)α)−1

(

εt

ηt

)

(32)

The code below estimates representation (31) by maximum likelihood:

// this program estimates the model in

// "The Demand for Money during Hyperinflations under Rational Expectations: I"

// by T. Sargent, IER 1977 using maximum likelihood

// variables are defined as follows:

// x=p_t-p_{t-1}, p being the log of the price level

// mu=m_t-m_{t-1}, m being the log of money supply

// note that in contrast to the paper eta and epsilon

// have variance 1 (they are multiplied by the standard deviations)

var x mu a1 a2; varexo epsilon eta; parameters alpha lambda sig_eta

sig_epsilon; lambda=.5921; alpha=-2.344; sig_eta=.001;

sig_epsilon=.001;

model;

x=x(-1)-lambda*a1(-1)+(1/(lambda+alpha*(1-lambda)))*sig_epsilon*epsilon

-(1/(lambda+alpha*(1-lambda)))*sig_eta*eta;

mu=(1-lambda)*x(-1)+lambda*mu(-1)-lambda*a2(-1)+(1+alpha*(1-lambda))/(lambda+alpha*(1-lambda))

*sig_epsilon*epsilon-(1-lambda)/(lambda+alpha*(1-lambda))*sig_eta*eta;

a1=(1/(lambda+alpha*(1-lambda)))*sig_epsilon*epsilon-(1/(lambda+alpha*(1-lambda)))*sig_eta*eta;

a2=(1+alpha*(1-lambda))/(lambda+alpha*(1-lambda))*sig_epsilon*epsilon

-(1-lambda)/(lambda+alpha*(1-lambda))*sig_eta*eta;

end; steady; shocks;

var eta; stderr 1; var epsilon; stderr 1;

end;

estimated_params;

// ML estimation setup

// parameter name, initial value, boundaries_low, ..._up;

lambda, .5, 0.25, 0.75; alpha, -2, -8, -0.1; sig_eta, .0001, 0.0001,

0.3; sig_epsilon, .0001, 0.0001, 0.3; end;

varobs mu x; unit_root_vars x;

estimation(datafile=cagan_data,first_obs=1,nobs=34,mh_replic=0,mode_compute=2,mode_check);

The data for the estimation are from Cagan (1956) for Germany1. The shape of the negative of the log-likelihood
function is plotted in figure 21.

1The data are demeaned, allowing us to ignore constant terms in the model

Figure 21: Negative of the Log-Likelihood function for the hyperinflation model

Parameter Estimate s.d. t-stat
λ 0.6617 0.0535 12.3754
α -4.7654 3.0689 1.5528
ση 0.2447 0.2170 1.1275
σε 0.0727 0.0100 7.2946

Table 6: Results from Maximum likelihood

Next we estimate this model using Bayesian techniques. For this we replace the last part of the code above with

estimated_params;

// Bayesian setup

lambda, uniform_pdf, 0.68, .5;

alpha, uniform_pdf, -5, 2;

sig_eta,uniform_pdf, .5, 0.25;

sig_epsilon, uniform_pdf, .5, 0.25;

end;

varobs mu x; unit_root_vars x;

estimation(datafile=cagan_data,nobs=34,mh_replic=25000,mh_nblocks=1,mh_jscale=1,mode_compute=4);

Note that the priors we have chosen are reasonably loose, but given that there are only 34 observations we have
opted to use somewhat informative priors. The posteriors and priors are plotted in figure 22.

Figure 22: Posteriors and priors for the hyperinflation model

7. Solving and estimating Hall’s permanent income model

This note describes a suite of Dynare programs to simulate and estimate the permanent income model of Hall(1988).

7.1 The model

A consumer faces the following decision problem:

max
{ct,kt}∞

t=0

−
1

2

∞
∑

t=0

βt (ct − bt)
2

subject to the following constraints

ct + kt = Rkt−1 + dt

dt = ρdt−1 + (1 − ρ)µd + εd

bt = ρbbt−1 + (1 − ρb)µb + εb

We assume that βR = 1 and that there is no depreciation, which leads to the following law of motion for
investment:

it = kt − kt−1

From the Lagrangian for this problem one obtains the following first order conditions:

µct = µct+1

µct = bt − ct

where µct is the marginal utility of consumption or the Lagrange multiplier.
Note that this problem has a unit root and so there exist no steady state. However, we can supply ONE solution

corresponding to an initial k that we give to Dynare.
The following Dynare code simulates data from this model:

periods 5000;

var c k mu_c b d in;

varexo e_d e_b;

parameters R rho rho_b mu_b mu_d; R=1.05;

rho = 0;

mu_b=30;

mu_d=5;

rho_b = 0;

model(linear);

c+k = R*k(-1) + d;

mu_c = b - c;

mu_c=mu_c(+1);

d= rho*d(-1)+ mu_d*(1-rho) + e_d;

b=(1-rho_b)*mu_b+rho_b*b(-1)+e_b;

in = k - k(-1);

end;

//With a unit root, there exists no steady state. Use the following trick.

// Supply ONE solution corresponding to the initial k that you named.

initval;

d=mu_d;

k=100;

c = (R-1)*k +d;

mu_c=mu_b-c;

b=mu_b;

end;

shocks;

var e_d;

stderr 1;

var e_b;

stderr 1;

end;

steady;

check;

stoch_simul(dr_algo=1, order=1, periods=500, irf=10);

save data_hall c in;

Next, we proceed to estimate a subset of the parameters of the model by ML using simulated data obtained from
the previous code. To do so we add the following few lines of code. Notice how we use the command unit root vars
to initialize the Kalman filter with a diffuse prior. We do this because k is not stationary – Hall’s model with βR = 1
has a solution for k that has a unit root and that is co-integrated with consumption.

estimated_params;

// ML estimation setup

// parameter name, initial value, boundaries_low, ..._up;

rho, -0.0159, -0.9, 0.9;

R, 1.0074, 0, 1.5;

end;

varobs c in;

// declare the unit root variables for diffuse filter

unit_root_vars k;

estimation(datafile=data_hall,first_obs=101,nobs=200,mh_replic=0,mode_compute=4,mode_check);

We proceed to estimate the same subset of the parameters using Bayesian techniques: This is implemented in
Dynare by adding the following few lines of code.

estimated_params;

rho, beta_pdf, .1, 0.2;

R, normal_pdf,1.02,0.05;

end;

varobs c in;

Figure 23: A shock to εb

// declare the unit root variables for diffuse filter

unit_root_vars k;

estimation(datafile=data_hall,first_obs=101,nobs=200,mh_replic=1000,mh_nblocks=2,mh_jscale=2);

8. Solving and estimating Ryoo and Rosen’s (2004) model of the mar-

ket for engineers

This section describes a suite of Dynare programs to simulate and estimate a model of Aloysius Siow (1984) and
Jaewoo Ryoo and Sherwin Rosen (2004) of the labor market for engineers.

8.1 The model

Ryoo and Rosen studied a partial equilibrium model determining a stock of ’engineers’ Nt; the number of new
entrants into engineering school, st; and the wage level wt of engineers.2 It takes k periods of schooling to become
an engineer. The model consists of the following components:
The present value of wages of an engineer:

Pt = (1 − δ)βPt+1 + (1 − δ)k βkwt+k

A supply of new engineering students:

st = a0 + a1Pt + est

2Sherwin Rosen showed this version of the model to Sargent in Chicago in fall 1995.

Figure 24: A shock to εd

A time-to-build structure of the education process:

Nt = (1 − δ)Nt−1 + st−k

A demand curve for engineers:
Nt = d0 − d1wt + edt

The following Dynare code simulates data from this model:

periods 500;

var s N P W;

varexo e_s e_d;

parameters a0 a1 delta d0 d1 bet k;

a0=10;

a1=1;

d0=1000;

d1=1;

bet=.99;

delta=.02;

model(linear);

s=a0+a1*P+e_s; // flow supply of new entrants

N=(1-delta)*N(-1) + s(-4); // evolution of the stock

N=d0-d1*W+e_d; // stock demand equation

P=bet*(1-delta)*P(+1)+bet^4*(1-delta)^4*W(+4); // present value of

//wages

Figure 25: priors and posteriors from the Bayesian estimation of Hall’s model

end;

initval;

s=0;

N=0;

P=0;

W=0;

end;

shocks;

var e_d;

stderr 1;

var e_s;

stderr 1;

end;

steady;

check;

stoch_simul(dr_algo=1, order=1, periods=500, irf=10);

save data_rosen s N P W;

Note that in this code we set k = 4. Dynare does not let you use a parameter as lag length.
Next, we estimate a subset of the parameters of the model by ML using simulated data obtained from the previous

code. To do so we add the following lines of code.

Figure 26: A shock to est

estimated_params;

a1, .5, -10, 10;

d1, .5, -20, 40;

stderr e_d, .5, .05, 5;

stderr e_s, .3, .05, 6; end;

// these are the ranges for the parameters

varobs W N;

estimation(datafile=data_rosen,first_obs=101,nobs=200,mh_replic=0,mode_compute=4,mode_check);

Figure plots the negative of the log-likelihood function for the parameters.
We proceed to estimate the same subset of the parameters using Bayesian techniques. We implemented this in

Dynare by adding the following lines of code.

estimated_params;

a1, gamma_pdf, .5, .5;

d1, gamma_pdf, 2, .5;

stderr e_d,inv_gamma_pdf,.5, 30;

stderr e_s,inv_gamma_pdf,.2, 30;

// The syntax for to input the priors is the following:

// variable name, prior distribution, parameters of distribution.

Figure 27: Negative of the Log-Likelihood for Ryoo and Rosen’s model

end;

varobs W N;

estimation(datafile=data_rosen,first_obs=101,nobs=200,mh_replic=10000,mh_nblocks=2,

mh_jscale=2,mode_compute=0,mode_file=rosen_estimateML2_mode);

9. Estimation of Bansal and Yaron(2004) consumption growth process

Let consumption growth have the state space representation

xt = ρxt−1 + σxεx,t

∆ct = µc + xt−1 + σcεc,t

where the shocks εc,t and εx,t are i.i.d. standard normal. We use quarterly growth rate of per capita U.S. consumption
over the period 1947.2-2003.3 to estimate the four unknown parameters. The following Dynare code does this by
maximum likelihood.

var x y;

varexo e_x e_u;

parameters rho sig_x sig_u mu_y;

rho = .98;

mu_y=.015;

Figure 28: priors and posteriors from the Bayesian estimation of Ryoo and Rosen’s model

sig_x=0.00025;

sig_u=.0078;

model(linear);

x=rho*x(-1) + sig_x*e_x;

y=mu_y + x(-1) +sig_u*e_u;

end;

initval;

x=0;

y=mu_y;

end;

steady;

shocks;

var e_x;

stderr 1;

var e_u;

stderr 1;

end;

estimated_params;

// ML estimation setup

// parameter name, initial value, boundaries_low, ..._up;

rho, 0, -0.999, 0.999; // use this for unconstrained max

likelihood

// rho, .980, .975, .999 ; // use this for long run risk model

// sig_x, .0004,.0001,.05 ; // use this for the long run risk model

sig_x, .0005, .00000000001, .01; // use this for unconstrained

max likelihood sig_u, .007,.001, .1; mu_y, .014, .0001, .04;

end;

varobs y;

estimation(datafile=data_consRicardoypg,first_obs=1,nobs=227,mh_replic=0,mode_compute=4,mode_check);

// mode_compute=4 uses Chris Sim’s csminwel algorithm to maximize the likelihood function.

// Using mode_compute=1 which uses the Matlab function fmincon or using mode_compute=2 which uses

// Lester Ingber’s Adaptive Simulated Annealing algorithm produce almost identical results. However,

// model_compute=3 which uses fminunc gives the wrong answer.

Table 7 reports the estimated parameter from Dynare.

Parameter Estimate s.d. t-stat
rho 0.5623 0.1927 2.9179
sigx 0.0030 0.0011 2.6921
sigu 0.0040 0.0007 5.4519
muy 0.0056 0.0005 10.7613

Table 7: Results from Maximum likelihood

Figure 29 also plots the negative of the log-likelihood function along several dimensions.
Next, we estimate the same model using Bayesian techniques and two different sets of priors. The first set of

priors are tight around the calibrated values used in Bansal and Yaron (2004). To accomplish this, we can simply
change the following lines of code from our MLE program above.

Figure 29: Negative of the Log-Likelihood function for the consumption growth model

Figure 30: Prior and Posteriors for the Bansal and Yaron (2004) endownment process

estimated_params;

rho, beta_pdf, .98, .01;

mu_y, uniform_pdf, .005, .0025;

sig_u, inv_gamma_pdf, .003, inf;

sig_x, inv_gamma_pdf, .003, inf;

// The syntax for to input the priors is the following:

// variable name, prior distribution, parameters of distribution.

end;

varobs y;

estimation(datafile=data_consRicardoypg,first_obs=1,nobs=227,mh_replic=5000,mh_nblocks=1,mh_jscale=1)

Figure 30 plots the prior and the posterior densities for the estimated parameters. The posteriors are plotted in
black and priors in gray. A vertical green line denotes a posterior mean.

Next, we set vague priors for ρ and for µy.

rho, beta_pdf, .5, .2;

mu_y, uniform_pdf, .005, .0025;

sig_u, inv_gamma_pdf, .003, inf;

Figure 31: Prior and Posteriors for the Bansal and Yaron (2004) endownment process

sig_x, inv_gamma_pdf, .003, inf;

Figure 31 plots the prior and the posterior densities for the estimated parameters. Notice that with a more diffuse
prior on ρ, the posteriors for µy are tighter.

10. Robust permanent income and pricing

Hansen, Sargent, and Tallarini (1999) study consumption, saving, and security market prices in a permanent in-
come model in which the agents mistrust their specification of the economic environment. The preferences of a
representative consumer are defined in terms of the following risk-sensitivity recursion:

Ut = ut + βRt(Ut+1) (33)

Rt(Ut+1) ≡
2

σ
log Et[exp(

σUt+1

2
)] (34)

with ut being the per-period utility function. Note that σ = 0 corresponds to Von Neumann-Morgenstern preferences.
The authors use a linear quadratic framework described below. Furthermore, they show that for a given (β, σ) pair
there exists another pair (β̃, 0) that gives the same equilibrium quantities. This observation simplifies solution and
estimation of the model significantly. For the σ = 0 case and HST’s linear-quadratic habit persistence model,
preferences become

U0 = E0

∞
∑

t=0

βt(−(st − b)2) (35)

with b being a subsistence point and s defined as

st = (1 + λ)ct − λht−1 (36)

ht = δhht−1 + (1 − δh)ct (37)

where ct is current period consumption and ht−1 the habit stock. There is a linear production technology:

ct + it = γkt−1 + dt (38)

kt = δkkt−1 + it (39)

dt = µd + d̂t + d̄t (40)

d̄t = (φ1 + φ2)d̄t−1 − φ1φ2d̄t−2 + cd̄ ∗ e1t (41)

d̂t = (α1 + α2)d̂t−1 − α1α2d̂t−2 + c
d̂
∗ e2t (42)

e1t and e1t are standard normal iid shocks. Finally, we define the gross return on capital as

R ≡ δk + γ (43)

and impose
Rβ = 1. (44)

To solve the planning problem, we formulate a Lagrangian with multipliers 2βtµst on (36), 2βtµht on (37) and 2βtµct

on (38) (after substituting for investment it). The first order necessary conditions are:

µst = b − st (45)

µct = (1 + λ)µst + (1 − δh)µht (46)

µht = βEt[δhµht+1 − λµst+1] (47)

µct = µct+1 (48)

plus (36)-(39). The following code estimates the model using maximum likelihood:

// Estimates the Hansen Sargent and Tallarini model by maximum likelihood.

var s c h k i d dhat dbar mus muc muh gamma R; varexo e_dhat e_dbar;

parameters lambda deltah deltak mud b bet phi1 phi2 cdbar alpha1

alpha2 cdhat;

bet=0.9971;

deltah=0.682;

lambda=2.443;

alpha1=0.813;

alpha2=0.189;

phi1=0.998;

phi2=0.704;

mud=13.710;

cdhat=0.155;

cdbar=0.108;

b=32;

deltak=0.975;

model(linear);

R=deltak+gamma;

R*bet=1;

s=(1+lambda)*c-lambda*h(-1);

h=deltah*h(-1)+(1-deltah)*c;

k=deltak*k(-1)+i;

c+i=gamma*k(-1)+d;

mus=b-s;

muc=(1+lambda)*mus+(1-deltah)*muh;

muh=bet*(deltah*muh(+1)-lambda*mus(+1));

muc=bet*R*muc(+1);

d=mud+dbar+dhat;

dbar=(phi1+phi2)*dbar(-1) - phi1*phi2*dbar(-2) +

cdbar*e_dbar;

dhat=(alpha1+alpha2)*dhat(-1) -

alpha1*alpha2*dhat(-2)+ cdhat*e_dhat;

end;

shocks;

var e_dhat;

stderr 1;

var e_dbar;

stderr 1;

end;

estimated_params;

bet, .91, .9, .99999;

deltah, 0.4, 0.1, 0.8;

lambda, 2, 0.1, 50;

alpha1, 0.8, 0.6, 0.99999;

alpha2, 0.2,0.01,0.5;

phi1, 0.8, 0.6, 0.99999;

phi2, 0.5, 0.3, 0.9;

mud, 10, 1,50;

cdhat, 0.1, 0.05, 0.2;

cdbar, 0.1, 0.05, 0.2;

end;

varobs c i;

estimation(datafile=dataHST,first_obs=1,nobs=500,mode_compute=4,mode_check);

To estimate the model using Bayesian techniques simply replace the last part of the code with

estimated_params;

bet,uniform_pdf, .9499999999, 0.0288675134306;

deltah,uniform_pdf, 0.45, 0.202072594216;

lambda,uniform_pdf,25.05,14.4048892163;

alpha1,uniform_pdf, 0.8, 0.115470053809;

alpha2,uniform_pdf, 0.25, 0.144337567297;

phi1,uniform_pdf,0.8,0.115470053809;

phi2,uniform_pdf, 0.5, 0.288675134595;

mud,uniform_pdf, 24.5, 14.1450815951;

cdhat,uniform_pdf,0.175,0.0721687836487;

cdbar,uniform_pdf, 0.175, 0.0721687836487;

Figure 32: negative of the log-likelihood for the robust permanent income model

end;

varobs c i;

estimation(datafile=dataHST,nobs=500,mode_check,mh_replic=10000,mh_nblocks=1,mh_jscale=0.2);

The sliced negative of the log-likelihood function is given by pictures (32) and (33). The maximization routine
used here (mode compute=4) is Chris Sims’ csminwel. Note that this procedure uses a random number generator,
so different runs of this code can lead to different estimates (the routine might get stuck in a local extremum). Thus,
it is important to run the code several times.

11. Asset pricing with risks for the long run

11.1 Introduction

Following Bansal and Yaron (2004), we specify dividend and consumption growth as containing a small but highly
persistent predictable component. These dynamics, when combined with Epstein and Zin (1989) preferences, help
explain financial phenomena including the equity premium puzzle. What allows this specification to be successful
is the ability of Epstein and Zin preferences to separate risk aversion from intertemporal elasticity of substitution,
possibly allowing both to be larger than one. Dynare enables us to compute policy functions for the response of
price-dividend ratios to an increase in the long-run predictable component of dividend and consumption growth.
Using the specification just described, this impact turns out to be positive. In economic terms, the intertemporal
substitution effect dominates the wealth effect. In response to higher expected growth rates, agents buy more assets
and this increases the wealth to consumption ratio. In the standard power utility (constant relative risk aversion)
model, the wealth effect dominates the substitution effect. Dynare shows that the policy functions for the latter

Figure 33: negative of the log-likelihood for the robust permanent income model

specification display the opposite sign for the feedback of the predictable component of growth on price-dividend
ratios. Hence the former specification helps explain high equity premia, because when the IES is larger than one,
agents fear that a reduction in economic growth prospects will lower asset prices.

11.2 Setup of the model

Specify preferences as follows:

Ut =

[

(1 − δ)C
1−γ

θ

t + δ
(

Et

[

U1−γ
t+1

])
1
θ

]

θ
1−γ

(49)

where

θ =
1 − γ

1 − 1/Ψ
(50)

Here γ reflects the degree of risk aversion, while Ψ take into account intertemporal elasticity of substitution. When
θ = 1, preferences collapse to the standard CRRA case. Hence this model can be regarded as a general one.

There are N assets in this economy, whose one period ahead return is denoted as Rj,t+1, ∀j = 1, ..., N . The
share of investment in asset j is denoted as ωj,t. Hence there are N − 1 linearly independent elements in ωt =
[

ω1,t ω2,t ... ωN,t

]′
, since portfolio shares must satisfy:

N
∑

j=1

ωj,t = 1, ∀t (51)

The representative agent is endowed with an initial stock of the consumption asset A0 that can be either consumed
or invested in the N assets in the competitive market. Therefore, her wealth evolves according to:

At+1 = (At − Ct)ω
′
tRt (52)

Following Epstein and Zin (1991), first order conditions to price an asset that entitles to a claim on a stream of
consumption good imply:

vc,t = Et

[

δθe[−
θ
Ψ

∆ct+1+(θ−1)rc,t+1](1 + vc,t+1)e
∆ct+1

]

(53)

where

∆ct = log(Ct) − log(Ct−1)

rc,t+1 = log Rc,t+1

Rc,t+1 =
(1 + vc,t+1)e

∆ct+1

vc,t

vc,t =
Pc,t

Ct

Similarly, first order condition to price an asset that entitles to a dividend Dt in each period imply:

vd,t = Et

[

δθe[−
θ
Ψ

∆ct+1+(θ−1)rc,t+1](1 + vd,t+1)e
∆dt+1

]

(54)

where

∆dt = log(Dt − Dt−1)

vd,t =
Pd,t

Dt

We close the model, by specifying exogenous processes for consumption and dividend growth:

∆ct = µc + xt−1 + εc,t (55)

∆dt = µd + λdxxt−1 + εd,t (56)

xt = ρxxt−1 + εx,t (57)

where the shocks εc,t, εd,t and εs,t are i.i.d. normal with mean zero and standard deviations σc, σd and σx, respectively.
For stationarity |ρx| < 1. The economy is completely described by equations (53)-(57).

11.3 Calibration and simulation

We calibrate the parameters of the model in two different ways. Our guidelines in doing this are:

• use a reasonable risk aversion

• model the process of x as highly persistent

One important feature that, following the lines of Bansal and Yaron (2004), we would like to display is the different
way in which xt affects price-dividend ratios in the presence of Epstein and Zin utility functions. As we stressed
before the channel through which a highly persistent and predictable component of consumption and dividend
growth combined with Epstein-Zin preferences helps explaining the equity premium puzzle is the possibility of
having intertemporal elasticity of substitution and risk aversion greater than one at the same time. Hence we
want to experiment both a calibration where this condition is met and a standard constant relative risk aversion
preferences environment, where risk aversion greater than one implies IES smaller than unity. The latter specification
is a particular case in our model, as it only requires to put θ = 1.

11.3.1 CRRA specification

We calibrate parameters as in Table 8. The numbers are the same as in Bansal and Yaron (2004) except for risk
IES, that is now the inverse of risk aversion.

Parameter Calibration
Ψ 0.133
θ 1
γ 7.5
δ 0.99
µc 0.0015
σc 0.0078
µd 0.0015
λdx 3
σd 0.0351
ρx 0.979
σx 0.0003432

Table 8: Parameters’ calibration. The value of θ is calibrated to satisfy equation (50).

The Dynare code reported below implements a first order approximation of the model.

periods 500;

var dc, dd, v_c, v_d, x; varexo e_c, e_x, e_d;

parameters DELTA THETA PSI MU_C MU_D RHO_X LAMBDA_DX;

DELTA=.99;

PSI=1.5;

THETA=(1-7.5)/(1-1/PSI);

MU_C=0.0015;

MU_D=0.0015;

RHO_X=.979;

LAMBDA_DX=3;

model;

v_c = DELTA^THETA *exp((-THETA/PSI)*dc(+1)+(THETA-1)*log((1+v_c(+1))*exp(dc(+1))/v_c))* (1+v_c(+1))*exp(dc(+1)); %

v_d = DELTA^THETA * exp((-THETA/PSI)*dc(+1) + (THETA-1)*log((1+v_c(+1))*exp(dc(+1))/v_c) ) * (1+v_d(+1))*exp(dd(+1)); %

dc = MU_C + x(-1) + e_c;

dd = MU_D + LAMBDA_DX*x(-1) + e_d; %

x = RHO_X * x(-1) + e_x;

end;

initval;

v_c=15;

v_d=15;

dc=MU_C;

dd=MU_D;

x=0;

e_c=0;

e_x=0;

e_d=0;

end;

shocks;

var e_c;

stderr .0078;

var e_x;

stderr .0078*.044;%

var e_d;

stderr .0078*4.5;

end;

steady(solve_algo=0); check;

stoch_simul(dr_algo=1, order=1, periods=1000, irf=30);

datasaver(’simudata’,[]);

Dynare computes policy and transition functions as reported in Table 9. Since these equations are expressed in
terms of xt−1, it is possible to express the system in terms of time t state variable using equation (57):

vc,t = 50.005− 8086.753(xt − x)

vd,t = 50.005− 5598.521(xt − x)

Equations (55)-(57) complete the description of the solution.

x ∆c ∆d vc vd

Constant 0 0.0015 0.0015 50.005844 50.005844
x (-1) 0.979 1 3 -7916.931499 -5480.952577
ec 0 1 0 0 0
ed 0 0 1 0 0
ex 1 0 0 -8086.753319 -5598.521529

Table 9: Policy and transition functions, CRRA calibration (Matlab output).

It is also interesting to look at the impulse response functions to a shock to xt. Figure 34 reports the response
of the variables of the system to a one standard deviation shock to x. According to the rationale discussed above,
both price-dividend and price-consumption ratios fall.

Figure 34: Impulse response function to a shock to xt (CRRA preferences).

11.3.2 Epstein-Zin specification

This section focuses on the more general Epstein-Zin setup. In particular, we allow both risk aversion and IES to be
larger than one. Ψ and, via equation (50), θ are the only two parameters to differ between Table 8 and Table 10.

Parameter Calibration
Ψ 1.5
θ −19.5
γ 7.5
δ 0.99
µc 0.0015
σc 0.0078
µd 0.0015
λdx 3
σd 0.0351
ρx 0.979
σx 0.0003432

Table 10: Parameters’ calibration. The value of θ is calibrated to satisfy equation (50).

By running the Dynare program with the change in the calibration of Ψ and θ, the policy functions for vc and
vd change to:

vc,t = 104.209 + 1146.215(xt − x)

vd,t = 104.209 + 8023.510(xt − x)

Therefore, a shock to xt will now have a positive effect on price-dividend ratios. Separating risk aversion from
intertemporal substitution and modelling both as greater than unity produces the effect of changing the sign of the
impact of the forecastable part of consumption and dividend growth on price-dividend ratios. The IRF reported in
Figure 35 stress this finding.

Figure 35: Impulse response function to a shock to xt (CRRA preferences).

11.4 Bayesian estimation

We perform a Bayesian estimation of the general model, using simulated series. The model has three shocks εc, εx

and εd and three observables vd, ∆c and ∆d. The priors that we use are reported in Table 11. A slight modification
to the program reported in the previous pages enables us to estimate the parameters of the model in Dynare. The
following code was been used:

var dc, dd, v_c, v_d, x;

varexo e_c, e_x, e_d;

parameters DELTA THETA PSI MU_C MU_D RHO_X LAMBDA_DX;

DELTA=.99;

PSI=1.5;

THETA=(1-7.5)/(1-1/PSI);

MU_C=0.0015;

MU_D=0.0015;

RHO_X=.979;

LAMBDA_DX=3;

model;

v_c = DELTA^THETA * exp((-THETA/PSI)*dc(+1) + (THETA-1)*log((1+v_c(+1))*exp(dc(+1))/v_c) ) * (1+v_c(+1))*exp(dc(+1)); %

v_d = DELTA^THETA * exp((-THETA/PSI)*dc(+1) + (THETA-1)*log((1+v_c(+1))*exp(dc(+1))/v_c) ) * (1+v_d(+1))*exp(dd(+1)); %

dc = MU_C + x(-1) + e_c; %

dd = MU_D + LAMBDA_DX*x(-1) + e_d; %

x = RHO_X * x(-1) + e_x;

end;

initval;

v_c=15;

v_d=15;

dc=MU_C;

dd=MU_D;

x=0;

e_c=0;

e_x=0;

e_d=0;

end;

shocks;

var e_d;

stderr .001;

var e_c;

stderr .001;

var e_x;

stderr .001;

end;

steady;

estimated_params;

DELTA, beta_pdf, 0.99,.001;

THETA,normal_pdf,-7.5, 0.005;

PSI,normal_pdf,1.5, 0.005;

MU_C,normal_pdf,0.001, 0.005;

MU_D,normal_pdf,0.001, 0.005;

RHO_X,normal_pdf,.98, 0.05;

LAMBDA_DX,normal_pdf,3, 0.05;%

stderr e_d,inv_gamma_pdf,.01, inf; %

stderr e_x,inv_gamma_pdf,.001, inf;

stderr e_c,inv_gamma_pdf,.0001, inf;

end;

varobs v_d dd dc;

estimation(datafile=simudata,mh_replic=5000);

In terms of Dynare’s output Figure 36 displays priors’ distributions, while Figure 37 reports a comparison between
prior’s and posterior’s distributions. Since the data generating process is exactly the model that we are estimating,
it is not very surprising that the distribution of the posterior has very little variance around the true parameters.

Parameter Prior Distribution Mean Std. Dev.
Ψ Normal 1.5 .005
θ Normal −19.5 .005
δ Beta 0.99 .001
µc Normal 0.0015 .005
σc Inv. Gamma 0.0078 inf
µd Normal 0.0015 .005
λdx Normal 3 .05
σd Inv. Gamma 0.0351 inf
ρx Normal 0.979 .05
σx Inv. Gamma 0.0003432 inf

Table 11: Priors.

Dynare also produces a table of summary statistics of the estimation. As already seen from the Figures, the mean
of the posterior distributions is almost coincidental with the correct values and the estimated confidence interval is
very tight around these values. Table 12 reports these numbers.

Figure 36: Priors.

Prior Mean Post. Mean Confidence Interval Prior dbn Post std dev
δ 0.99 0.9896 0.988 0.991 Beta 0.001
θ -7.5 -7.501 -7.5102 - 7.4934 Normal 0.005
Ψ 1.5 1.4981 1.4898 1.5049 Normal 0.005
µc 0.001 0.0012 0.0001 0.0021 Normal 0.005
µd 0.001 0.0017 -0.0004 0.0044 Normal 0.005
ρx 0.98 0.9786 0.9753 0.9813 Normal 0.05
λdx 3 2.9963 2.9205 3.0812 Normal 0.05
σd 0.01 0.0368 0.0355 0.038 Inv. Gamma Inf
σx 0.001 0.0003 0.0003 0.0003 Inv. Gamma Inf
σc 0 0.0076 0.0074 0.0079 Inv. Gamma Inf

Table 12: Posterior.

Figure 37: Posterior.

A. List of Dynare Files

This appendix lists the dynare files used in this paper. All files and the data used are contained in the file examples.zip.
In case simulated data was used in the estimation one first needs to run the relevant dynare code to simulate the
data.

• GrowthApproximate.mod - solves the neoclassical growth model and simulates data

• GrowthEstimate.mod - estimates the neoclassical growth model

• TwocountryApprox.mod - solves the two country production economy and simulates data

• TwocountryEstim.mod - estimates the two country production economy

• Fig11XX.mod - solves a deterministic growth model and reproduces the graph XX in chapter 11 of Ljungqvist
and Sargent (2004)

• Figv3 11XX.mod - solves a deterministic growth model and reproduces the graph XX in chapter 11 of Ljungqvist
and Sargent (20XX)

• sargent77.mod - solves the model in Sargent (1977)

• sargent77ML.mod, sargent77Bayes.mod - estimate the model in Sargent (1977) using maximum likelihood and
Bayesian MCMC methods, respectively

• hall1.mod - solves the model in Hall (1988) and simulates data

• hall1estimateML.mod, hall1estimateBayes.mod - estimate the model in Hall (1988) using maximum likelihood
and Bayesian MCMC methods, respectively

• rosen.mod - solves the model in Ryoo and Rosen (2004) and simulates data

• rosenestimateML.mod, rosenestimateBayes.mod - estimate the model in Ryoo and Rosen (2004) using maxi-
mum likelihood and Bayesian MCMC methods, respectively

• BansalYaronML.mod, BansalYaronBayes.mod - estimate the consumption-growth process of Bansal and Yaron
(2004) using maximum likelihood and Bayesian MCMC methods, respectively

• HSTML.mod, HSTBayes.mod - estimate the model in Hansen, Sargent, and Tallarini (1999) using maximum
likelihood and Bayesian MCMC methods, respectively

• AssetPricingApproximation.mod - solves the model of Bansal and Yaron (2004) and simulates data

• AssetPricingEstimate.mod - estimates the model of Bansal and Yaron (2004)

